Metodo dei Trapezi

Enrico Bertolazzi

Introduzione

Data la funzione $f(x)$ si approssima

$$\int_a^b f(x)\,dx = \frac{h\,(f(a)+f(b))}{2} + h\left(\sum_{i=1}^{n-1} f(x_i)\right) - \frac{(b-a)\,h^2\,D(D(f))(z)}{12}$$

dove $h = \frac{b-a}{n}$ e $x_i = a + i\,h$

Carica le librerie

```
> restart ;
  with(plots):
Warning, the name changecoords has been redefined
```

Definisce la procedura Trapezi

```
> trapezi := proc(f,a,b,n)
    local i, xi, h, res ;
    h   := evalf((b-a)/n) ;
    res := evalf((f(b)+f(a))/2,20) ;
    for i from 1 to n-1 do
      xi  := a + i * h ;
      res := evalf(res + f(xi), 20) ;
    end do;
    return res * h ;
  end proc :
```

Procedura di Stampa

```
> trapezi_print := proc(f,a,b,n)
    local i, xi, h, pts, GA, GB, GC ;
    h   := evalf((b-a)/n) ;
    pts := [] ;
    for i from 0 to n do
      xi  := a + i * h ;
      pts := [ op(pts), [xi, f(xi)] ] ;
    end do:
```

```
    GA := plot(pts,style=line,color=yellow,filled=true):
    GB := plot(pts,style=line,thickness=2,color=red):
    GC := plot(f,a..b,thickness=2,color=blue):
    display({GA,GB,GC},axes=normal,title='Trapezi');
  end proc :
```

Esempio d'uso

```
> # Definisce la funzione da approssimare
  f := x -> x*sin(x) ;
```

$$f := x \to x \sin(x)$$

```
> a := 0 ;
  b := 10 ;
  n := 10 ;
```

$$a := 0$$

$$b := 10$$

$$n := 10$$

```
> # risolve il problema
  trapezi(f,a,b,n) ;
```

$$7.087834496$$

```
> # disegna la procedura ``trapezi''
  trapezi_print(f,a,b,n) ;
```

Trapezi



Esercizio: stimare il numero di intervalli per il metodo dei trapezi affinche' l'errore sia inferiore a 10^(-6)

```
> # calcolo la derivata seconda della funzione
  f2 := D(D(f)) ;
```

$$f2 := x \to 2\cos(x) - x\sin(x)$$

```
> # discegno la derivata nell'intervallo considerato e stimo a vista la
  sua maggiorazione
  plot(f2,a..b) ;
```




```
> # la maggiorazione pu essere |f2(x)| < 10
  f2max := 10 ;
```

```
> # la stima dell'errore sara quindi
  stima := (b-a)*h^2*f2max/12 = 10^(-6) ;
```

$$stima := \frac{5}{6} h^2 \, f2max = \frac{1}{1000000}$$

```
> # sostituisco h=(b-a)/N ;
  stimaN := subs(h=(b-a)/N,stima) ;
```

$$stimaN := \frac{250 \, f2max}{3 N^2} = \frac{1}{1000000}$$

```
> # ricavo N^2 dalla stima
  N2 := isolate(stimaN,N*N) ;
```

$$N2 := N^2 = \frac{250000000}{3}\ f2max$$

```
> # stimo gli intervalli
  evalf(sqrt(rhs(N2)),6) ;
```

$$9128.73\ \sqrt{f2max}$$

```
> # gli intervalli saranno quindi 28868.
>
```